
AIMMS Language Reference - Set, Set Element and String Expressions

This file contains only one chapter of the book. For a free download of the complete book in pdf format, please visit www.aimms.com.

Copyright © 1993–2018 by AIMMS B.V. All rights reserved.

AIMMS B.V.
Diakenhuisweg 29-35
2033 AP Haarlem
The Netherlands
Tel.: +31 23 5511512

AIMMS Inc.
11711 SE 8th Street
Suite 303
Bellevue, WA 98005
USA
Tel.: +1 425 458 4024

AIMMS Pte. Ltd.
55 Market Street #10-00
Singapore 048941
Tel.: +65 6521 2827

AIMMS
SOHO Fuxing Plaza No.388
Building D-71, Level 3
Madang Road, Huangpu District
Shanghai 200025
China
Tel.: ++86 21 5309 8733

Email: info@aimms.com
WWW: www.aimms.com

AIMMS is a registered trademark of AIMMS B.V. IBM ILOG CPLEX and CPLEX is a registered trademark of IBM Corporation. GUROBI is a registered trademark of Gurobi Optimization, Inc. KNITRO is a registered trademark of Artelys. WINDOWS and EXCEL are registered trademarks of Microsoft Corporation. \TeX , \LaTeX , and $\AMS-\LaTeX$ are trademarks of the American Mathematical Society. LUCIDA is a registered trademark of Bigelow & Holmes Inc. ACROBAT is a registered trademark of Adobe Systems Inc. Other brands and their products are trademarks of their respective holders.

Information in this document is subject to change without notice and does not represent a commitment on the part of AIMMS B.V. The software described in this document is furnished under a license agreement and may only be used and copied in accordance with the terms of the agreement. The documentation may not, in whole or in part, be copied, photocopied, reproduced, translated, or reduced to any electronic medium or machine-readable form without prior consent, in writing, from AIMMS B.V.

AIMMS B.V. makes no representation or warranty with respect to the adequacy of this documentation or the programs which it describes for any particular purpose or with respect to its adequacy to produce any particular result. In no event shall AIMMS B.V., its employees, its contractors or the authors of this documentation be liable for special, direct, indirect or consequential damages, losses, costs, charges, claims, demands, or claims for lost profits, fees or expenses of any nature or kind.

In addition to the foregoing, users should recognize that all complex software systems and their documentation contain errors and omissions. The authors, AIMMS B.V. and its employees, and its contractors shall not be responsible under any circumstances for providing information or corrections to errors and omissions discovered at any time in this book or the software it describes, whether or not they are aware of the errors or omissions. The authors, AIMMS B.V. and its employees, and its contractors do not recommend the use of the software described in this book for applications in which errors or omissions could threaten life, injury or significant loss.

This documentation was typeset by AIMMS B.V. using \LaTeX and the LUCIDA font family.

Chapter 5

Set, Set Element and String Expressions

Expressions are organized arrangements of operators, constants, sets, indices, parameters, and variables that evaluate to either a set, a set element, a numerical value, a logical value, a string value, or a unit value. Expressions form the core of the AIMMS language. In the previous chapters you already have seen some elementary examples of expressions.

Several types of expressions

In this chapter, set, set element and string expressions are presented in detail. For expressions that evaluate to either numerical or logical values, you are referred to Chapter 6. Expressions that evaluate to unit values are discussed in Section 32.6

This chapter

5.1 Set expressions

Set expressions play an important role in the construction of index domains of indexed identifiers, as well as in constructing the domain of execution of particular indexed statements. The AIMMS language offers a powerful set of set expressions, allowing you to express complex set constructs in a clear and concise manner.

Set expressions

A set expression is evaluated to yield the value of a set. As with all expressions in AIMMS, set expressions come in two forms, *constant* and *symbolic*. Constant set expressions refer to explicit set elements directly, and are mainly intended for set initialization. The tabular format of set initialization is treated in Section 28.2.



Constant set expressions

Symbolic set expressions are formulas that can be executed to result in a set. The contents of this set can vary throughout the execution of your model depending on the values of the other model identifiers referenced inside the symbolic formulas. Symbolic set expressions are typically used for specifying index domains. In this section various forms of set expressions will be treated.

Symbolic set expressions

set-primary :

Syntax

*set-expression :*

The simplest form of set expression is the reference to a set. The reference can be scalar or indexed, and evaluates to the current contents of that set.

Set references

Through the `.Domain` suffix you have access to the current contents of the index domain of any indexed identifier. You can use the `.Domain` suffix as if it were a set identifier.

The .Domain suffix

Consider the indexed variable `Transport(i, j)` defined over (a subset of) the set `Cities × Cities`. You can use the reference

Example

```
Transport.Domain
```

as a two-dimensional subset of `Cities × Cities`, containing all tuples that currently lie within the `IndexDomain` of the variable `Transport(i, j)`.

5.1.1 Enumerated sets

An *enumerated* set is a set defined by an explicit enumeration of its elements. Such an enumeration includes literal elements, set element expressions, and (constant or symbolic) element ranges. An enumerated set can be either a simple or a compound set. If you use an *integer element range*, an integer set will result.

Enumerated sets

Enumerated sets come in two flavors: *constant* and *symbolic*. Constant enumerated sets are preceded by the keyword DATA, and must only contain literal set elements. These set elements do not have to be contained in single quotes unless they contain characters other than the alpha-numeric characters, the underscore, the plus or the minus sign.

*Constant
enumerated sets*

The following simple and compound set assignments illustrate constant enumerated set expressions.

Example

```
Cities := DATA { Amsterdam, Rotterdam, 'The Hague', London, Paris, Berlin, Madrid } ;

DutchRoutes := DATA { (Amsterdam, Rotterdam ), (Amsterdam, 'The Hague'),
                       (Rotterdam, Amsterdam ), (Rotterdam, 'The Hague') } ;
```

Any enumerated set not preceded by the keyword DATA is considered symbolic. Symbolic enumerated sets can also contain element parameters. In order to distinguish between literal set elements and element parameters, all literal elements inside symbolic enumerated sets must be quoted.

*Symbolic
enumerated sets*

The following two set assignments illustrate the use of enumerated sets that depend on the value of the element parameters SmallestCity, LargestCity and AirportCity.

Examples

```
ExtremeCities := { SmallestCity, LargestCity } ;

Routes        := { (LargestCity, SmallestCity), (AirportCity, LargestCity ) } ;
```

The following two set assignments contrast the semantics between constant and symbolic enumerated sets.


```
SillyExtremes := DATA { SmallestCity, LargestCity } ;
! contents equals { 'SmallestCity', 'LargestCity' }

ExtremeCities := { SmallestCity, LargestCity, 'Amsterdam' };
! contents equals e.g. { 'The Hague', 'London', 'Amsterdam' }
```


The syntax of enumerated set expressions is as follows.

enumerated-set :


Syntax



element-tuple :



tuple-component :



All elements in an enumerated set must have the same dimension.

By using the `..` operator, you can specify an *element range*. An element range is a sequence of consecutively numbered elements. The following set assignments illustrate both constant and symbolic element ranges. Their difference is explained below.

Element range

```


NodeSet      := DATA { node1 .. node100 } ;
FirstNode    := 1;
LastNode     := 100;

IntegerNodes := { FirstNode .. LastNode } ;
    
```


The syntax of element ranges is as follows.

element-range :

Syntax



range-bound :



A range bound must consists of an integer number, and can be preceded or followed by a common prefix or postfix string, respectively. The prefix and postfix strings used in the lower and upper range bounds must coincide.

Prefix and postfix strings

If you use an element range in a static enumerated set expression (i.e. preceded by the keyword `DATA`), the range can only refer to explicitly numbered elements, which need not be quoted. By padding the numbered elements with zeroes, you indicate that AIMMS should create all elements with the same element length.

Constant range

As the begin and end elements of a constant element range are literal elements, you cannot use a constant element range to create sets with dynamically changing border elements. If you want to accomplish this, you should use the `ElementRange` function, which is explained in detail in Section 5.1.4. Its use in the following example is self-explanatory. The following set assignments illustrate a constant element range and its equivalent formulation using the `ElementRange` function.

```
NodeSet      := DATA { node1 .. node100 } ;
PaddedNodes := DATA { node001 .. node100 } ;

NodeSet      := ElementRange( 1, 100, prefix: "node", fill: 0 ) ;
PaddedNodes := ElementRange( 1, 100, prefix: "node", fill: 1 ) ;
```

*Constant range
versus
ElementRange*

Element ranges in a symbolic enumerated set can be used to create integer ranges. Now, both bounds can be numerical expressions. Such a construct will result in the *dynamic* creation of a number of *integer* elements based on the value of the numerical expressions at the range bounds. Such integer element ranges can only be assigned to *integer* sets (see Section 3.2.2). An example of a dynamic integer range follows.

```
IntegerNodes := { FirstNode .. LastNode } ;
```

*Symbolic
integer range*

In this example `IntegerNodes` must be an integer set.

If the elements in the range are not consecutive but lie at regular intervals from one another, you can indicate this by adding a `BY` modifier with the proper interval length. For static enumerated sets the interval length must be a constant, for dynamic enumerated sets it can be any numerical expression. The following set assignments illustrate a constant and symbolic element range with nonconsecutive elements.

```
EvenNodes      := DATA { node2 .. node100 by 2 } ;

StepSize       := 2 ;
EvenIntegerNodes := { FirstNode .. LastNode by StepSize } ;
```

*Nonconsecutive
range*

When specifying element tuples in an enumerated set expression, it is possible to create multiple tuples in a concise manner using cross products. You can specify multiple elements for a particular tuple component in the cross product either by grouping single elements using the `[and]` operators or by using an element range, as shown below.

```
DutchRoutes := DATA { ( Amsterdam, [Rotterdam, 'The Hague'] ),
                      ( Rotterdam, [Amsterdam, 'The Hague'] ) } ;
! creates { ( 'Amsterdam', 'Rotterdam' ), ( 'Amsterdam', 'The Hague' ),
!          ( 'Rotterdam', 'Amsterdam' ), ( 'Rotterdam', 'The Hague' ) }

Network      := DATA { ( node1 .. node100, node1 .. node100 ) } ;
```

Element tuples

The assignment to the set Network will create a set with 10,000 elements.

5.1.2 Constructed sets

A *constructed set* expression is one in which the selection of elements is constructed through filtering on the basis of a particular condition. When a constructed set expression contains an index, AIMMS will consider the resulting tuples for every element in the binding set.

Constructed sets

The following set assignments illustrate some constructed set expressions, assuming that *i* and *j* are indices into the set Cities.

Example


```

LargeCities := { i | Population(i) > 500000 } ;
Routes := { (i,j) | Distance(i,j) } ;
RoutesFromLargestCity := { (LargestCity, j) in Routes } ;
    
```

In the latter assignment route tuples are constructed from LargestCity (an element-valued parameter) to every city *j*, where additionally each created tuple is required to lie in the set Routes.

constructed-set :

Syntax




binding-domain :



binding-tuple :

binding-element :



The tuple selection in a constructed set expression behaves exactly the same as the tuple selection on the left-hand side of an assignment to an indexed parameter. This means that all tuple components can be either an explicit quoted set element, a general set element expression, or a binding index. The tuple can be subject to a logical condition, further restricting the number of elements constructed.

Binding domain

5.1.3 Set operators

There are four binary set operators in AIMMS: Cartesian product, intersection, union, and difference. Their notation and precedence are given in Table 5.1. Expressions containing these set operators are read from left to right and the operands can be any set expression. There are no unary set operators.

Four set operators

Operator	Notation	Precedence
intersection	*	3 (high)
difference	-	2
union	+	2
Cartesian product	CROSS	1 (low)

Table 5.1: Set operators

The following set assignments to integer sets and Cartesian products of integer sets illustrate the use of all available set operators.

Example

```

S := {1,2,3,4} * {3,4,5,6} ;      ! Intersection of integer sets: {3,4}.

S := {1,2} + {3,4} ;             ! Union of simple sets:
S := {1,3,4} + {2} + {1,2} ;    ! {1,2,3,4}

S := {1,2,3,4} - {2,4,5,7} ;    ! Difference of integer sets: {1,3}.

T := {1,2} cross {1,2} ;        ! The cross of two integer sets:
                                ! {(1,1),(1,2),(2,1),(2,2)}.

U := {(1,2),(1,3)} cross {4,5} ; ! The cross of a compound and integer set:
                                ! {(1,2,4),(1,2,5),(1,3,4),(1,3,5)}.

```

The precedence and associativity of the operators is demonstrated by the assignments

```

T := A cross B - C ;           ! Same as A cross (B - C).
T := A - B * C + D ;          ! Same as (A - (B * C)) + D.
T := A - B * C + D * E ;      ! Same as (A - (B * C)) + (D * E).

```

The operands of union, difference, and intersection must have the same dimensions.

```

T := {(1,2),(1,3)} * {(1,3)} ;      ! Same as {(1,3)}.

T := {(1,2),(1,3)} + {(i,j) | a(i,j) > 1} ; ! Union of enumerated
                                                ! and constructed set of
                                                ! the same dimension.

T := {(1,2),(1,3)} + {(1,2,3)} ;    ! ERROR: dimensions differ.

```

5.1.4 Set functions

A special type of set expression is a call to one of the following set-valued functions *Set functions*

- ElementRange,
- SubRange,
- ConstraintVariables,
- VariableConstraints, or
- A user-defined function.

The ElementRange and SubRange functions are discussed in this section, while the functions ConstraintVariables and VariableConstraints are discussed in Section 15.1. The syntax of and use of tags in function calls is discussed in Section 10.2.

The ElementRange function allows you to *dynamically* create or change the contents of a set of non-integer elements based on the value of integer-valued scalars expressions. *The function ElementRange*

The ElementRange function has two mandatory integer arguments. *Arguments*

- *first*, the integer value for which the first element must be created, and
- *last*, the integer value for which the last element must be created.

In addition, it allows the following four optional arguments.

- *incr*, the integer-valued interval length between two consecutive elements (default value 1),
- *prefix*, the prefix string for every element (by default, the empty string),
- *postfix*, the postfix string (by default, the empty string), and
- *fill*, a logical indicator (0 or 1) whether the numbers must be padded with zeroes (default value 1).

If you use any of the optional arguments you must use their formal argument names as tags.

Consider the sets S and T initialized by the constant set expressions

Example

```
NodeSet      := DATA { node1 .. node100 } ;
PaddedNodes := DATA { node001 .. node100 } ;
EvenNodes    := DATA { node2 .. node100 by 2 } ;
```

These sets can also be created in a dynamic manner by the following applications of the ElementRange function.

```
NodeSet      := ElementRange( 1, 100, prefix: "node", fill: 0 );
PaddedNodes := ElementRange( 1, 100, prefix: "node", fill: 1 );
EvenNodes    := ElementRange( 2, 100, prefix: "node", fill: 0, incr: 2 );
```

The SubRange function has three arguments:

- a simple or compound *set*,
- the *first* element, and
- the *last* element.

The SubRange function

The result of the function is the subset ranging from the *first* to the *last* element. If the first element is positioned after the last element, the empty set will result.

Assume that the set Cities is organized such that all foreign cities are consecutive, and that FirstForeignCity and LastForeignCity are element-valued parameters into the set Cities. Then the following assignment will create the subset ForeignCities of Cities

Example

```
ForeignCities := SubRange( Cities, FirstForeignCity, LastForeignCity ) ;
```

5.1.5 Iterative set operators

Iterative operators form an important class of operators that are especially designed for indexed expressions in AIMMS. There are set, element-valued, arithmetic, statistical, and logical iterative operators. The syntax is always similar.

Iterative operators

iterative-expression :



Syntax

The first argument of all iterative operators is a *binding domain*. It consists of a single index or tuple of indices, optionally qualified by a logical condition. The second argument and further arguments must be expressions. These expressions are evaluated for every index or tuple in the binding domain, and the result is input for the particular iterative operator at hand. Indices in the expressions that are not part of the binding domain of the iterative operators are referred to as *outer indices*, and must be bound elsewhere.

Explanation

AIMMS possesses the following set-related iterative operators:

- the Sort operator for sorting the elements in a domain,
- the NBest operator for obtaining the *n* best elements in a domain according to a certain criterion, and
- the Intersection and Union operators for repeated intersection or union of indexed sets.

Set-related iterative operators

Sorting the elements of a set is a useful tool for controlling the flow of execution and for presenting reordered data in the graphical user interface. There are two mechanisms available to you for sorting set elements

Reordering your data

- the `OrderBy` attribute of a set, and
- the `Sort` operator.

The second and further operands of the `Sort` operator must be numerical, element-valued or string expressions. The result of the `Sort` operator will consist of precisely those elements that satisfy the domain condition, sorted according to the single or multiple ordering criteria specified by the second and further operands. Section 3.2 discusses the expressions that can be used for specifying an ordering principle.

Sorting semantics

Note that the set to which the result of the `Sort` operator is assigned must have the `OrderBy` attribute set to `User` (see also Section 3.2.1) for the operation to be useful. Without this setting AIMMS will store the elements of the result set of the `Sort` operator, but will discard the underlying ordering.

Receiving set

The following assignments will result in the same set orderings as in the example of the `OrderBy` attribute in Section 3.2.

Example

```
LexicographicSupplyCities := Sort( i in SupplyCities, i );
ReverseLexicographicSupplyCities := Sort( i in SupplyCities, -i );
SupplyCitiesByIncreasingTransport :=
  Sort( i in SupplyCities, Sum( j, Transport(i,j) );
SupplyCitiesByDecreasingTransportThenLexicographic :=
  Sort( i in SupplyCities, - Sum( j, Transport(i,j) ), i );
```

AIMMS will even allow you to sort the elements of a root set. Because the entire execution system of AIMMS is built around a fixed ordering of the root sets, sorting root sets may influence the overall execution in a negative manner. Section 13.2.7 explains the efficiency considerations regarding root set ordering in more detail.

Sorting root sets

You can use the `NBest` operator, when you need the n best elements in a set according to a single ordering criterion. The syntax of the `NBest` is similar to that of the `Sort` operator. The first expression after the binding domain is the criterion with respect to which you want elements in the binding domain to be ordered. The second expression refers to the number of elements n in which you are interested.

Obtaining the n best elements

The following assignment will, for every city *i*, select the three cities to which the largest transports emanating from *i* take place. The result is stored in the indexed set `LargestTransportCities(i)`.

Example

```
LargestTransportCities(i) := NBest( j, Transport(i,j), 3 );
```

With the `Intersection` and `Union` operators you can perform repeated set intersection or union respectively. A typical application is to take the repeated intersection or union of all instances of an indexed set. However, any set valued expression can be used on the second argument.

Repeated intersection and union

Consider the following indexed set declarations.

Example

```
Set IndSet1 {
  IndexDomain : s1;
  SubsetOf    : S;
}
Set IndSet2 {
  IndexDomain : s1;
  SubsetOf    : S;
}
```

With these declarations, the following assignments illustrate valid uses of the `Union` and `Intersection` operators.

```
SubS := Union( s1, IndSet1(s1) );
SubS := Intersection( s1, IndSet1(s1) + IndSet2(s1) );
```

5.1.6 Set element expressions as singleton sets

Element expressions can be used in a set expression as well. In the context of a set expression, AIMMS will interpret an element expression as the singleton set containing only the element represented by the element expression. Set element expressions are discussed in full detail in Section 5.2.

Element expressions ...

Using an element expression as a set expression can equivalently be expressed as a symbolic enumerated set containing the element expression as its sole element. Whenever there is no need to group multiple elements, AIMMS allows you to omit the surrounding braces.

... versus enumerated sets

The following set assignment illustrate some simple set element expressions used as a singleton set expression.

Example

```
! Remove LargestCity from the set of Cities
Cities -= LargestCity ;
```

```

! Remove first element from the set of Cities
Cities -= Element(Cities,1) ;

! Remove LargestCity and SmallestCity from Cities
Cities -= LargestCity + SmallestCity ;

! The set of Cities minus the CapitalCity
NonCapitalCities := Cities - CapitalCity ;

```

5.2 Set element expressions

Set element expressions reference a particular element or element tuple model from a set or a tuple domain. Set element expressions allow for *sliced assignment*—executing an assignment only for a lesser-dimensional subdomain by fixing certain dimensions to a specific set element. Potentially, this may lead to a vast reduction in execution times for time-consuming calculations.

Use of set element expressions

The most elementary form of a set element expression is an element parameter, which turns out to be a useful device for communicating set element information with the graphical interface. You can instruct AIMMS to locate the position in a table or other object where an end-user made changes to a numerical value, and have AIMMS pass the corresponding set element(s) to an element parameter. As a result, you can execute data input checks defined over these element parameters, thereby limiting the amount of computation. This issue is discussed in more detail in the help regarding the Identifier Selection dialog.


Passing elements from the GUI

AIMMS supports several types of set element expressions, including *references* to parameters and (bound) indices, *lag-lead-expressions*, element-valued *functions*, and *iterative-expressions*. The last category turns out to be a useful device for computing the proper value of element parameters in your model.

Element expressions

element-expression :

Syntax



The format of list expressions are the same for element and numerical expressions. They are discussed in Section 6.1.2.


5.2.1 Element references

An element reference is any reference to either an element parameter or a (bound) index. An element reference can be followed by a tag or tuple of tags as a way to refer to a component or subtuple of components.

Element references

element-reference :

Syntax



You can use tags when you want to refer to a subtuple of an element of a compound set. The tags must be declared as attributes in the corresponding compound root set. Tags and their use are explained in detail in Section 3.2.3.

Use of tags

Assume that `orig` and `dest` (in this order!) are tags in the compound set `Routes`, and that the element parameter `MainRoute` contains a two-tuple from the compound set `Cities`×`Cities`. The following statement computes the reverse route using a tagged element reference.

Example

```
ReverseRoute := MainRoute.(dest,orig) ;
```

5.2.2 Intrinsic functions for sets and set elements

AIMMS supports functions to obtain the position of an element within a set, the cardinality (i.e. number of elements) of a set, the n -th element in a set, the element in a non-compatible set with the identical string representation, and the numerical value represented by a set element. If S is a set identifier, i an index bound to S , l an element, and n a positive integer, then possible calls to the `Ord`, `Card`, `Element`, `ElementCast` and `Val` functions are given in Table 5.2.

The element-related functions...

The `Ord`, `Card` and `Element` functions can be applied to both simple and compound sets. In fact you can even apply `Card` to parameters and variables—it simply returns the number of nondefault elements associated with a certain data structure.

... for simple and compound sets

Function	Value	Meaning
Ord(<i>i</i>)	<i>integer</i>	Ordinal, returns the relative position of the index <i>i</i> in the set <i>S</i> . Does <i>not</i> bind <i>i</i> .
Ord(<i>l</i> , <i>S</i>)	<i>integer</i>	Returns the relative position of the element <i>l</i> in set <i>S</i> . Returns zero if <i>l</i> is not an element of <i>S</i> .
Card(<i>S</i>)	<i>integer</i>	Cardinality of set <i>S</i> .
Element(<i>S</i> , <i>n</i>)	<i>element</i>	Returns the element in set <i>S</i> at relative position <i>n</i> . Returns the empty element tuple if <i>S</i> contains less than <i>n</i> elements.
ElementCast(<i>S</i> , <i>l</i>)	<i>element</i>	Returns the element in set <i>S</i> , which corresponds to the textual representation of an element <i>l</i> in any other index set.
Val(<i>l</i>)	<i>numerical</i>	Returns the numerical value represented by <i>l</i> , or a runtime error if <i>l</i> cannot be interpreted as a number
Max(<i>e</i> ₁ , ..., <i>e</i> _{<i>n</i>})	<i>Max</i>	Returns the set element with the highest ordinal
Min(<i>e</i> ₁ , ..., <i>e</i> _{<i>n</i>})	<i>Min</i>	Returns the set element with the lowest ordinal

Table 5.2: Intrinsic functions operating on sets and set elements

By default, AIMMS does not allow you to use indices associated with one root set hierarchy in your model, in references to index domains associated with another root set hierarchy of your model. The function `ElementCast` allows you to cross root set boundaries, by returning the set element in the root set associated with the first (set) argument that has the identical name as the element (in another root set) passed as the second argument. The function `ElementCast` has an optional third argument *create* (values 0 or 1, with a default of 0), through which you can indicate whether you want elements which cannot be cast to the indicated set must be created within that set. In this case, a call to `ElementCast` will never fail. You can find more information about root sets, as well as an illustrative example of the use of `ElementCast`, in Section 9.1.

Crossing root set boundaries

In this example, we again use the set `Cities` initialized through the statement

Example

```
Cities := DATA { Amsterdam, Rotterdam, 'The Hague', London, Paris, Berlin, Madrid } ;
```

The following table illustrates the intrinsic element-valued functions.

If your model contains a set with elements that represent numerical values, you cannot directly use such elements as a numerical value in numerical expressions, unless the set is an integer set (see Section 3.2.2). To obtain the

The Val function

Expression	Result
Ord('Amsterdam', Cities)	1
Ord('New York', Cities)	0 (i.e. not in the set)
Card(Cities)	7
Element(Cities, 1)	'Amsterdam'
Element(Cities, 8)	' ' (i.e. no 8-th element)

numerical value of such set elements, you can use the Val function. You can also apply the Val function to strings that represent a numerical value. In both cases, a runtime error will occur if the element or string argument of the Val function cannot be interpreted as a numerical value.

The element-valued Min and Max functions operate on two or more element-valued expressions *in the same (sub-)set hierarchy*. If the arguments are references to element parameters (or bound indices), then the Range attributes of these element parameters or indices must be sets in a single set hierarchy. Through these functions you can obtain the elements with the lowest and highest ordinal relative to the set equal to highest ranking range set in the subset hierarchy of all its arguments. If one or more of the arguments are explicit labels, then AIMMS will verify that these labels are contained in that set, or will return an error otherwise. A compiler error will result, if no such set can be determined (i.e., when the function call refers to explicit labels only).

The Min and Max functions

The Tuple function has a variable number of input arguments, all of which must be element expressions. The value of this function is a literal element of a compound set. The dimension of the created tuple should match the dimension of the context in which it is used, and AIMMS must be able to determine the compound set from which the presented tuple is supposed to be an element from the context. When the result of the Tuple function is assigned to an element parameter, no assignment is made when the result is not an element of the range set of the parameter.

The Tuple function

The following assignment creates a reference to a particular route, namely between the cities represented by the element parameters HarborCity and AirportCity.

Example

```
MainRoute := Tuple( AirportCity, HarborCity );
```

You can also assign a tuple with literal elements.

```
MainRoute := Tuple( 'Amsterdam', 'Rotterdam' );
```

In both cases the created tuple should be an element of the range set Routes of the parameter MainRoute.

5.2.3 Element-valued iterative expressions

AIMMS offers special iterative operators that let you select a specific element from a simple or compound domain. Table 5.3 shows all such operators that result in a set element value. The syntax of iterative operators is explained in Section 5.1.5. The second column in this table refers to the required number of expression arguments following the binding domain argument.

Selecting elements

Name	# Expr.	Computes for all elements in the domain
First	0	the first element (tuple)
Last	0	the last element (tuple)
Nth	1	the n -th element (tuple)
Min	1	the value of the element expression for which the expression reaches its minimum ordinal value
Max	1	the value of the element expression for which the expression reaches its maximum ordinal value
ArgMin	1	the first element (tuple) for which the expression reaches its minimum value
ArgMax	1	the first element (tuple) for which the expression reaches its maximum value

Table 5.3: Element-valued iterative operators

The binding domain of the First, Last, Nth, Min, Max, ArgMin, and ArgMax operator can only consist of a single index in either a simple or compound set, and the result is a single element in that domain. You can use this result directly for indexing or referencing an indexed parameter or variable. Alternatively, you can assign it to an element parameter in the appropriate domain.

Single index

The ArgMin and ArgMax operators return the element for which an expression reaches its minimum or maximum value. The allowed expressions are:

Compared expressions

- numerical expressions, in which case AIMMS performs a numerical comparison,
- string expressions, in which case AIMMS uses the normal alphabetic ordering, and
- element expressions, in which case AIMMS compares the ordinal numbers of the resulting elements.

For element expressions, the iterative Min and Max operators return expression *values* with the minimum and maximum ordinal value.

The following assignments illustrate the use of some of the domain related iterative operators. The identifiers on the left are all element parameters. *Example*

```

FirstNonSupplyCity := First ( i | not Exists(j | Transport(i,j)) ) ;
SecondSupplyCity  := Nth  ( i | Exists(j | Transport(i,j)), 2 ) ;
SmallestSupplyCity := ArgMin( i, Sum(j, Transport(i,j)) ) ;
LargestTransportRoute := ArgMax( r, Transport(r) ) ;

```

Note that the iterative operators `Exists` and `Sum` are used here for illustrative purposes, and are not set- or element-related. They are treated in Sections 6.2.5 and 6.1.6, respectively.

5.2.4 Lag and lead element operators

There are four binary element operators, namely the lag and lead operators `+`, `++`, `-` and `--`. The first operand of each of these operators must be an element reference (such as an index or element parameter), while the second operand must be an integer numerical expression. There are no unary element operators. *Lag and lead operators...*

Lag and lead operators are used to relate an index or element parameter to preceding and subsequent elements in a set. Such correspondence is well-defined, except when a request extends beyond the bounds of the set. *... explained*

There are two kinds of lag and lead operators, namely *noncircular* and *circular* operators which behave differently when pushed beyond the beginning and the end of a set. *Noncircular versus circular*

- The noncircular operators (`+` and `-`) consider the ordered set elements as a *sequence* with no elements before the first element or after the last element.
- The circular operators (`++` and `--`) consider ordered set elements as a *circular chain*, in which the first and last elements are linked.

Let S be a set, i a set element expression, and k an integer-valued expression. The lag and lead operators `+`, `++`, `-`, `--` return the element of S as defined in Table 5.4. Please note that these operators are also available in the form of `+=`, `-=`, `++=` and `--=`. The operators in this form can be used in statements like: *Definition*

```

CurrentCity := 'Amsterdam';
CurrentCity --= 1; ! Equal to CurrentCity := CurrentCity -- 1;

```

Lag/lead expr.	Meaning
$i + k$	The element of S positioned k elements after i ; the empty element if there is no such element.
$i ++ k$	The circular version of $i + k$.
$i - k$	The member of S positioned k elements before i ; the empty element if there is no such element.
$i -- k$	The circular version of $i - k$.

Table 5.4: Lag and lead operators

You cannot always use lag and lead operators in combination with literal set elements. The reason for this is clear: a literal element can be an element of more than one set, and in general, unless the context in which the lag or lead operator is used dictates a particular (domain) set, it is impossible for AIMMS to determine which set to work with.

Not for literal elements

Lag and lead operators are frequently used in indexed parameters and variables, and may appear on the left- and right-hand side of assignments. You should be careful to check the correct use of the lag and lead operators to avoid making conceptual errors. For more specific information on the lag and lead operators refer to Section 8.2, which treats assignments to parameters and variables.

Verify the effect of lags and leads

Consider the set `Cities` initialized through the assignment

Example

```
Cities := DATA { Amsterdam, Rotterdam, 'The Hague', London, Paris, Berlin, Madrid } ;
```

Assuming that the index i and the element parameter `CurrentCity` both currently refer to 'Rotterdam', Table 5.5 illustrates the results of various lag/lead expressions.

Lag/lead expression	Result
$i+1$	'The Hague'
$i+6$	''
$i++6$	'Amsterdam'
$i++7$	'Rotterdam'
$i-2$	''
$i--2$	'Madrid'
<code>CurrentCity+2</code>	'London'
<code>'Rotterdam' + 1</code>	ERROR

Table 5.5: Example of lag and lead operators

5.3 String expressions

String expressions are useful for


- creating descriptive texts associated with particular set elements and identifiers, or
- forming customized messages for display in the graphical user interface or in output reports.

String expressions

This section discusses all available string expressions in AIMMS.

string-expression :

Syntax



The format of list expressions are the same for string-valued and numerical expressions. They are discussed in Section [6.1.2](#).

5.3.1 String operators

There are three binary string operators in AIMMS, string concatenation (+ operator), string subtraction (- operator), and string repetition (* operator). There are no unary string operators.

String operators

The simplest form of composing strings in AIMMS is by the concatenation of two existing strings. String concatenation is represented as a simple addition of strings by means of the + operator.

String concatenation

In addition to string concatenation, AIMMS also supports subtraction of two strings by means of the - operator. The result of the operation $s_1 - s_2$ where s_1 and s_2 are string expressions will be the substring of s_1 obtained by

String subtraction

- omitting s_2 on the right of s_1 when s_1 ends in the string s_2 , or
- just s_1 otherwise.

You can use the multiplication operator `*` to obtain the string that is the result of a given number of repetitions of a string. The left-hand operand of the repetition operator `*` must be a string expression, while the right-hand operand must be an integer numerical expression.

String repetition

The following examples illustrate some basic string manipulations in AIMMS.

Examples

```
"This is " + "a string"      ! "This is a string"
"Filename.txt" - ".txt"      ! "Filename"
"Filename" - ".txt"          ! "Filename"
"--" * 5                      ! "-----"
```

5.3.2 Formatting strings

With the `FormatString` function you can compose a string that is built up from combinations of numbers, strings and set elements. Its arguments are:

*The function
FormatString*

- a *format string*, which specifies how the string is composed, and
- one or more *arguments* (number, string or element) which are used to form the string as specified.

The first argument of the function `FormatString` is a mixture of ordinary text plus *conversion specifiers* for each of the subsequent arguments. A conversion specifier is a code to indicate that data of a specified type is to be inserted as text. Each conversion specifier starts with the `%` character followed by a letter indicating its type. The conversion specifier for every argument type are given in Table 5.6.

*The format
string*

Conversion specifiers	Argument type
<code>%s</code>	String expression
<code>%e</code>	Element expression
<code>%f</code>	Floating point number
<code>%g</code>	Exponential format number
<code>%i</code>	Integer expression
<code>%n</code>	Numerical expression
<code>%u</code>	Unit expression
<code>%%</code>	<code>%</code> sign

Table 5.6: Conversion codes for the `FormatString` function

In the example below, the current value of the parameter `SmallVal` and `LargeVal` are 10 and 20, the current value of `CapitalCity` is the element 'Amsterdam', and `UnitPar` is a unit-valued parameter with value kton/hr. The following calls to `FormatString` illustrate its use.

Example

```
FormatString("The numbers %i and %i", 10, 20)           ! "The numbers 10 and 20"
FormatString("The numbers %i and %i", SmallVal, LargeVal) ! "The numbers 10 and 20"
FormatString("The number %n", 4*ArcTan(1))             ! "The number 3.14156"
FormatString("The string %s", "is printed")           ! "The string is printed"
FormatString("The element %e", CapitalCity)           ! "The element Amsterdam"
FormatString("The unit is %u", UnitPar)               ! "The unit is kton/hr"
```

By default, AIMMS will use a default representation for arguments of each type. By modifying the conversion specifier, you further dictate the manner in which a particular argument of the `FormatString` function is printed. *This is done by inserting modification flags in between the %-sign and the conversion character.* The following modification directives can be added:

Modification flags

- *flags*:
 - < for left alignment
 - <> for centered alignment
 - > for right alignment
 - + add a plus sign (nonnegative numbers)
 - ⌵ add a space (instead of the above + sign)
 - 0 fill with zeroes (right-aligned numbers only)
 - t print number using thousand separators, using local convention for both the thousand separator and decimal separator. Controlling these separators is via the options `Number 1000 separator` and `Number decimal separator`.
- *field width*: the converted argument will be printed in a field of at least this width, or wider if necessary
- *dot*: separating the field width from the precision
- *precision*: the number of decimals for numbers, or the maximal number of characters for strings or set elements.

It is important to note that the modification flags must be inserted in the order as described above.

Note the order

Both the field width and precision of a conversion specifier can be either an integer constant, or a wildcard, `*`. In the latter case the `FormatString` expects one additional integer argument for each wildcard just before the argument of the associated conversion specifier. This allows you to compute and specify either the field width or precision in a dynamic manner.

Field width and precision

When using the %n conversion specifier for a floating point number, AIMMS chooses whether to use a floating point or exponential format based on the global option `put_number_style`. Through the %f or %g conversion specifiers you can force the use of a floating point or exponential format, respectively.

Floating point vs. exponential format

The following calls to `FormatString` illustrate the use of modification flags.

Example

```
FormatString("The number %>+08i", 10)           ! "The number +0000010"
FormatString("The number %>t8i", 100000)        ! "The number 100,000"
FormatString("The number %> 8.2n", 4*ArcTan(1)) ! "The number 3.14"
FormatString("The number %> *.2n", 8,2,4*ArcTan(1)) ! "The number 3.14"
FormatString("The element %<5e", CapitalCity)   ! "The element Amsterdam"
FormatString("The element %<>5.3e", CapitalCity) ! "The element Ams "
```

AIMMS offers a number of special characters to allow you to use the full range of characters in composing strings. These special characters are contained in Table 5.7.

Special characters

Special character	text code	Meaning
\f	FF	Form feed
\t	HT	Horizontal tab
\n	LF	Newline character
\"	"	Double quote
\\	\	Backslash
\n	n	character n ($001 \leq n \leq 65535$)

Table 5.7: Special characters

Examples of the use of special characters within `FormatString` follow.

Example

```
FormatString("%i \037 \t %i %%", 10, 11) ! "10 % 11 %"
FormatString("This is a \"%s\" ", "string") ! "This is a "string" "
```

With the functions `StringToUpper`, `StringToLower` and `StringCapitalize` you can convert the case of a string to upper case, to lower case, or capitalize it, as illustrated in the following example.

Case conversion functions

```
StringToUpper("Convert to upper case") ! "CONVERT TO UPPER CASE"
StringToLower("CONVERT to lower case") ! "convert to lower case"
StringCapitalize("capitaLIZED SENTENCE") ! "Capitalized sentence"
```

5.3.3 String manipulation

In addition to the `FormatString` function, AIMMS offers a number of other functions for string manipulation. They are:

- `SubString` to obtain a substring of a particular string,
- `StringLength` to determine the length of a particular string,
- `FindString` to obtain the position of the first occurrence of a particular substring,
- `FindNthString` to obtain the position of the n -th occurrence of a particular substring, and
- `StringOccurrences` to obtain the number of occurrences of a particular substring.

*Other string
related
functions*

With the `SubString` function you can obtain a substring from a particular begin position m to an end position n (or to the end of the string if the requested end position exceeds the total string length). The positions m and n can both be negative (but with $m \leq n$), in which case AIMMS will start counting backwards from the end of the string. Examples are:

*The function
SubString*

```
SubString("Take a substring of me", 8, 16)    ! returns "substring"
SubString("Take a substring of me", 18, 100) ! returns "of me"
SubString("Take a substring of me", -5, -1)  ! returns "of me"
```

The function `StringLength` can be used to determine the length of a string in AIMMS. The function will return 0 for an empty string, and the total number of characters for a nonempty string. An example follows.

*The function
StringLength*

```
StringLength("Guess my length")             ! returns 15
```

With the functions `FindString` and `FindNthString` you can determine the position of the second argument, the *key*, within the first argument, the *search* string. The functions return zero if the key is not contained in the search string. The function `FindString` returns the position of the first occurrence of the key in the search string starting from the left, while the function `FindNthString` will return the position of the n -th appearance of the key. If n is negative, the function `FindNthString` will search backwards starting from the right. Examples are:

*The functions
FindString and
FindNthString*

```
FindString  ("Find a string in a string", "string"  ) ! returns 8
FindNthString ("Find a string in a string", "string", 2 ) ! returns 20
FindNthString ("Find a string in a string", "string", -1 ) ! returns 20

FindString  ("Find a string in a string", "this string") ! returns 0
FindNthString ("Find a string in a string", "string", 3 ) ! returns 0
```

By default, the functions `FindString` and `FindNthString` will use a case sensitive string comparison when searching for the key. You can modify this behavior through the option `Case_Sensitive_String_Comparison`.

Case sensitivity

The function `StringOccurrences` allows you to determine the number of occurrences of the second argument, the key, within the first argument, the *search* string. You can use this function, for instance, to delimit the number of calls to the function `FindNthString` a priori. An example follows.

The function StringOccurrences

```
StringOccurrences("Find a string in a string", "string" ) ! returns 2
```

5.3.4 Converting strings to set elements

Converting strings to new elements to or renaming existing elements in a set is not an uncommon action when end-users of your application are entering new element interactively or when you are obtaining strings (to be used as set elements) from other applications through external procedures. AIMMS offers the following support for dealing with such situations:

Converting strings to set elements

- the procedure `SetElementAdd` to add a new element to a set,
- the procedure `SetElementRename` to rename an existing element in a set, and
- the function `StringToElement` to convert strings to set elements.

The procedure `SetElementAdd` lets you add new elements to a set. Its arguments are:

Adding new set elements

- the *set* to which you want to add the new element,
- an *element parameter* into *set* which holds the new element after addition, and
- the *stringname* of the new element to be added.

When you apply `SetElementAdd` to a root set, the element will be added to that root set. When you apply it to a subset, the element will be added to the subset as well as to all its supersets, up to and including its associated root set.

Through the procedure `SetElementRename` you can provide a new name for an existing element in a particular set whenever this is necessary in your application. Its arguments are:

Renaming set elements

- the *set* which contains the element to be renamed,
- the *element* to be renamed, and
- the *stringname* to which the element should be renamed.

After renaming the element, all data defined over the old element name will be available under the new element name.

With the function `StringToElement` you can convert string arguments into (existing) elements of a set. If there is no such element, the function evaluates to the empty element. Its arguments are:

*The function
StringToElement*

- the *set* from which the element corresponding to *stringname* must be returned,
- the *stringname* for which you want to retrieve the corresponding element, and
- the optional *create* argument (values 0 or 1, with a default of 0) indicating whether nonexistent elements must be added to the set.

With the *create* argument set to 1, a call to `StringToElement` will always return an element in *set*. Alternatively to setting the *create* argument to 1, you can call the procedure `SetElementAdd` to add the element to the set.

The following example illustrates the combined use of `StringToElement` and `SetElementAdd`. It checks for the existence of the string parameter `CityString` in the set `Cities`, and adds it if necessary.

Example

```
ThisCity := StringToElement( Cities, CityString );
if ( not ThisCity ) then
  SetElementAdd( Cities, ThisCity, CityString );
endif;
```

Alternatively, you can combine both statements by setting the optional *create* argument of the function `StringToElement` to 1.

```
ThisCity := StringToElement( Cities, CityString, create: 1 );
```

Reversely, you can use the `%e` specifier in the `FormatString` function to get a pure textual representation of a set element, as illustrated in the following assignment.

*Converting
element to
string*

```
CityString := FormatString("%e", ThisCity );
```